/*
  Lars 20th gpsdo trial 140212
  in this revision : changed swedish comments to english
  
  This program is ONLY for 328!! 168 has to little SRAM for all variables
  tested with Arduino UNO and Arduino release 0022 (not 1.0)

  Gain set to 6500 for OCXO 10ppb range 
  still 1 ns res/ 1 us TIC / 8+8bit PWM DAC
  1PPS to interrupt (D2) and HC4046 pin 14
  1MHz from HC390 (div2*5)to HC4046 pin 3
  5 MHz from HC390 (div2)to timer1 (D5)
  1N5711+3.7k in series from HC4046 pin 15 to ADC0. 1nF NPO + 1M to ground on ADC0
  16bit PWM DAC with two-pole lowpass filter from D3 (39k+4.7uF+39k+4.7uF) and d11 (10M)
  Four dipswithches  from +5V via 100k,200k,390k resp. 820k to ADC3 
  same for ADC4 and both ADC 3 and 4 has 12kohm to ground
  One dipswitch connected to D7
  A recommendation is to connect one dipswitch from reset to a 10uf capacitor connected to ground.
  with this switch shorted the arduino will not reset every time the serial monitor is started. For
  downloading a new program the switch need to be set to open.

 */

#include <EEPROM.h>

long dacValue; // this is also same as "DACvalueOld" Note: is "0-65535" * timeconst
long dacValueOut = 32768; // 16bit PWM-DAC setvalue=startvalue Max 65535 (if not anything stored in the EEPROM)
long dacValueWithTempComp; // Note: is "0-65535" * timeconst
long FastLockDacValue; // filtered dacvalue for fast locking (settling)
volatile int TIC_Value; // analog read 0  - time value. About 1ns per bit with 3.7kohm+1nF
long TIC_ValueFiltered; // prefiltered TIC value
long TIC_ValueOld;
long TIC_ValueFilteredForPPS_lock; // prefiltered value just for PPS lock
//int TIC_Resolution = 1000; //in pS (* 1E-12 sec) to be used for gain calculation
//long VCO_MaxSpan =10000; // in ppt (* 1E-12 sec/sec) to be used for gain calculation
int tempADC1; // analog read 1  - int NTC
int tempADC2; // analog read 2  - ext temp eg LM35 or NTC - now used for temp compensation
long tempADC2_Filtered; // analog read 2  - ext temp filtered for temp correction
int modeADC3; // analog read 3  - used to set different modes like normal, hold etc
int timeADC4; // analog read 4  - used to set 16 different time constants
int spanADC5; // analog read 5  - used to set span of VCXO - not used yet
int modeNo;
int timeNo;
//int spanNo;
int modeNoOld;
int timeNoOld;
//int spanNoOld;
int tempCoeff = 0; // 6.5 = 1x10-11/c if 1°C= +10bit temp and 65dac bits is 1x10-11
int tempRef = 280; // offset for temp correction calculation - 280 is about 30C for LM35
long filterConst = 25; // pre-filter time const in secs (TIC-filtering)
const int filterDiv = 4; // filterConst = timeConst / filterDiv
long timeConst = 100; // Time constant in sec. Max 32767
long timeConstOld; // old Time constant in sec. Max 32767
int damping = 200; //Damping 100=1.0 50=0.5 etc
int TIC_Offset = 500; // ADC value for Reference time
long gain = 6500; //VCO freq DAC bits per TIC bit (1nS/bit 10ppb DACrange gives 6553.6) (1ppb range set to 65536)
const int PWMhighPin = 3; // high PWM-DAC pin (connected to 39k)
const int PWMlowPin = 11; // low PWM-DAC pin (connected to 10M)
int valuePWMhigh; // high PWM-DAC byte
int valuePWMlow; // low PWM-DAC byte
const int ppsLockedLED = 13; // LED pin for pps locked
const int extraSwitch = 7; // DIP Switch not used now but read
int valueExtraSwitch; // not used yet
int valueExtraSwitchOld; // not used yet (even in program)
unsigned long time; //tenths seconds since start
unsigned long timeOld; //old tenths seconds since start
int missedPPS = 0; // to be incremented every time pps is missed
volatile int PPS_ReadFlag = 0; // to be set every time pps is received
int lockPPSlimit = 100; //if TIC filtered for PPS within +- this for lockPSfactor * timeConst = PPSlocked
int lockPPSfactor = 5; // see above
unsigned long lockPPScounter = 0; // counter for PPSlocked
int PPSlocked = 0; //digital pin and prints 0 or 1
int i = 0; // counter for 300secs before storing temp and dac readings average
int j = 0; // counter for stored 300sec readings
int k = 0; // counter for stored 3hour readings
unsigned int StoreTIC_A[144]; //300sec storage
unsigned int StoreTempA[144];
unsigned int StoreDAC_A[144];
long sumTIC;
long sumTIC2;
long sumTemp;
long sumTemp2;
unsigned long sumDAC; 
unsigned long sumDAC2;
volatile unsigned int timer1CounterValue; 
volatile unsigned int overflowCount; // counter for timer1 overflows
volatile unsigned long timerCounts; // total count timer1+overflows
unsigned long timerCountsOld;
long timerCountsDiff;
long timerCountsAcc;
long timer_us; // timer1 value in microseconds offset from 1pps


ISR (TIMER1_OVF_vect) // Timer1 overflow interrupt
{
  ++overflowCount;               // count number of Counter1 overflows  
}  // end of TIMER1_OVF_vect


void pps() // interrupt routine
{
  timer1CounterValue = TCNT1;  // grab counter value before it changes any more
    // Note:due to Timer1 overflow interrupt this may be delayed and give wrong value (up to 3us delayed TNCT1 shows 30-31 instead of 15-29)
  if (TIFR1 & TOV1)   // if just missed an overflow. May still miss an overflow but corrected next time
     {
        overflowCount++;
     }
  timerCounts = (overflowCount * 65536 ) + timer1CounterValue;   // calculate total count - each overflow is 65536 more

  TIC_Value = analogRead(A0);

  PPS_ReadFlag = 1;
}
  
void calculation()
{

  timerCountsDiff = timerCounts - timerCountsOld;

  time = time + (timerCountsDiff + 250000) / 500000; // need to add 250000 to avoid truncation errors
  
  while (timerCountsDiff > 2500000)
    {
      timerCountsDiff = timerCountsDiff - 5000000;
    } 
  if (abs(timerCountsDiff + 65536) < 100) // corrects if missed overflow. not nice but..
    {
      timerCountsDiff = timerCountsDiff + 65536; 
      timerCounts = timerCounts + 65536;
    }
  
  timerCountsAcc = timerCountsAcc - timerCountsDiff;
  
  if (abs(timerCountsAcc) > 100) // reset if more than +-20uS. Should maybe add reset at start and maybe no reset in hold mode??
    {
      timerCountsAcc = (TIC_Value + 500)/200; // Not good to just check one sample. Should probably have some averaging.
    }
    
  timer_us = (timerCountsAcc - TIC_Value / 200) / 5;
  
  if (time - timeOld >11) // may give error after about 5000days (13 years) without reset!
    {
      missedPPS = missedPPS + 1;
    }
    
  timerCountsOld = timerCounts;  
  timeOld = time;
  
  TIC_ValueFilteredForPPS_lock = TIC_ValueFilteredForPPS_lock + (TIC_Value * 16 - TIC_ValueFilteredForPPS_lock) / 16; // Low Pass Filter for PPS lock
  lockPPScounter = lockPPScounter + 1;
  if (TIC_ValueFilteredForPPS_lock / 16 < TIC_Offset - lockPPSlimit)
    {
      lockPPScounter = 0;
    }
   if (TIC_ValueFilteredForPPS_lock / 16 > TIC_Offset + lockPPSlimit)
    {
      lockPPScounter = 0;
    }  
   if (lockPPScounter > timeConst * lockPPSfactor)
     {
      PPSlocked = 1;
     } 
     else
     {
      PPSlocked = 0;
     }
     
  digitalWrite(ppsLockedLED,PPSlocked); // turn on LED 13 if "locked"
  
  tempADC1 = analogRead(A1);
  tempADC2 = analogRead(A2);
  modeADC3 = analogRead(A3);
  timeADC4 = analogRead(A4);
  spanADC5 = analogRead(A5);
  valueExtraSwitch = digitalRead(extraSwitch); // not used so far
  
  if(modeADC3 <= 33){ modeNo = 0;} // mode 0-7 is normal regulation modes
  if(modeADC3 > 33) { modeNo = 1;} // mode 1-3 is fast etc see below
  if(modeADC3 > 104){ modeNo = 2;} 
  if(modeADC3 > 173){ modeNo = 3;}
  if(modeADC3 > 235){ modeNo = 4;}
  if(modeADC3 > 295){ modeNo = 5;}
  if(modeADC3 > 358){ modeNo = 6;}
  if(modeADC3 > 420){ modeNo = 7;}
  if(modeADC3 > 476){ modeNo = 8;} // mode 8-15 is hold of DAC value or sweep
  if(modeADC3 > 530){ modeNo = 9;} //  set mid DAC = 32768
  if(modeADC3 > 586){ modeNo = 10;} // set min DAC = 0
  if(modeADC3 > 642){ modeNo = 11;} // set max DAC = 65535
  if(modeADC3 > 692){ modeNo = 12;}
  if(modeADC3 > 740){ modeNo = 13;}
  if(modeADC3 > 792){ modeNo = 14;}
  if(modeADC3 > 843){ modeNo = 15;}
  
  if(modeNo != modeNoOld)// is this needed or not?
    {
    }
  modeNoOld = modeNo;   
  
  if(timeADC4 <= 33){ timeNo = 0; timeConst = 100;} // timeNo not use yet can probably be taken away
  if(timeADC4 > 33) { timeNo = 1; timeConst = 150;}
  if(timeADC4 > 104){ timeNo = 2; timeConst = 200;}
  if(timeADC4 > 173){ timeNo = 3; timeConst = 350;}
  if(timeADC4 > 235){ timeNo = 4; timeConst = 500;}
  if(timeADC4 > 295){ timeNo = 5; timeConst = 700;}
  if(timeADC4 > 358){ timeNo = 6; timeConst = 1000;}
  if(timeADC4 > 420){ timeNo = 7; timeConst = 1500;}
  if(timeADC4 > 476){ timeNo = 8; timeConst = 2000;}
  if(timeADC4 > 530){ timeNo = 9; timeConst = 3500;}
  if(timeADC4 > 586){ timeNo = 10; timeConst = 5000;}
  if(timeADC4 > 642){ timeNo = 11; timeConst = 7000;}
  if(timeADC4 > 692){ timeNo = 12; timeConst = 10000;}
  if(timeADC4 > 740){ timeNo = 13; timeConst = 15000;}
  if(timeADC4 > 792){ timeNo = 14; timeConst = 20000;}
  if(timeADC4 > 843){ timeNo = 15; timeConst = 30000;}
  if(modeNo ==  1){ timeNo = 16; timeConst = 50;} // fast used to quickly tune to correct TIC_value (risk for noise!!!!)
  if(modeNo ==  2){ timeNo = 17; timeConst = 25;} // fastfast
  if(modeNo ==  3){ timeNo = 18; timeConst = 12;} // fastfastfast
  
 /*
  if(lockPPScounter < 240 ) {timeConst = 50;} // start with 50sec Tc for fast settling of TIC
  if(lockPPScounter == 240 )  // 240 chosen to be slighly less than timeConst * lockPPSfactor (50*5)
   {
     dacValue = FastLockDacValue; // Use filtered value to get away with some of the ripple due to 50sec Tc
   }
 */
 
  filterConst =timeConst / filterDiv;
  
  if(timeConst != timeConstOld)
    {
  TIC_ValueOld = TIC_ValueOld / timeConstOld * timeConst; // recalculation of value
  TIC_ValueFiltered = TIC_ValueFiltered / timeConstOld * timeConst;
  dacValue = dacValue / timeConstOld * timeConst;
  FastLockDacValue = FastLockDacValue / timeConstOld * timeConst;
   }
  timeConstOld = timeConst;   
  
  
  if (time > 105 && modeADC3 < 476)// Don't change value the first 10 seconds or if in hold mode (Note: time is tenths of secs)
   {
  
   TIC_ValueFiltered = TIC_ValueFiltered + (TIC_Value * filterConst - TIC_ValueFiltered) / filterConst; // Low Pass Filter

   dacValue = dacValue + ((TIC_ValueFiltered - TIC_Offset * filterConst) * 100 / filterConst * gain / damping / timeConst);// Corr for time 
    // check calculation for truncation errors
 
   dacValue = dacValue + ((TIC_ValueFiltered - TIC_ValueOld) * 100 / filterConst * gain / 100 );// Corr for frequency
    // check calculation for truncation errors
    
   FastLockDacValue = FastLockDacValue  + (dacValue - FastLockDacValue ) / 100; // Low Pass Filter
   }
  
  tempADC2_Filtered = tempADC2_Filtered + (tempADC2 * 100 - tempADC2_Filtered) / 100; // Low Pass Filter
  
  dacValueWithTempComp = dacValue + ((tempRef * 100 - tempADC2_Filtered) * tempCoeff / 100 * timeConst ); // Corr for temp drift
  
  
  if (dacValue < 0)  // Constrain (x, a, b) can be used here
   { dacValue = 0;}
  if (dacValue > (65535 * timeConst))
   { dacValue = (65535 * timeConst);}

  TIC_ValueOld = TIC_ValueFiltered;
  
  dacValueOut = dacValueWithTempComp / timeConst; // PWM-DAC value
  if (dacValueOut < 0)   // Constrain (x, a, b) can be used here
   { dacValueOut = 0;}
  if (dacValueOut > 65535)
   { dacValueOut = 65535;}
   
  if(modeNo ==  9){ dacValueOut = 32768;} // set mid
  if(modeNo ==  10){ dacValueOut = 0;} // set min
  if(modeNo ==  11){ dacValueOut = 65535;} // set max

  valuePWMhigh = highByte(dacValueOut);
  valuePWMlow = lowByte(dacValueOut);
  analogWrite(PWMhighPin,valuePWMhigh);
  analogWrite(PWMlowPin,valuePWMlow);


/*
  Storage of average readings that is later printed
*/
  
  sumTIC = sumTIC + (TIC_Value * 10);
  sumTemp = sumTemp + (tempADC2 * 10);
  sumDAC = sumDAC + dacValueOut;
  i = i + 1;
  if (i == 300) // 300sec
  {
    StoreTIC_A[j]= sumTIC / i;
    sumTIC2 = sumTIC2 + sumTIC / i;
    sumTIC = 0;
    StoreTempA[j]= sumTemp / i;
    sumTemp2 = sumTemp2 + sumTemp / i;
    sumTemp = 0;
    StoreDAC_A[j]= sumDAC / i;
    sumDAC2 = sumDAC2 + sumDAC / i;
    sumDAC = 0;
    i = 0;
    j = j + 1;
    if (j % 36 == 0) // store every 36 x 300sec (3 hours)
    {
      sumTIC2 = sumTIC2 / 36;
      EEPROM.write(k, highByte(sumTIC2));
      EEPROM.write(k+144, lowByte(sumTIC2));
      sumTIC2 = 0;
      sumTemp2= sumTemp2 / 36;
      EEPROM.write(k+576, highByte(sumTemp2));
      EEPROM.write(k+720, lowByte(sumTemp2));
      sumTemp2 = 0;
      sumDAC2 = sumDAC2 / 36;
      EEPROM.write(k+288, highByte(sumDAC2));
      EEPROM.write(k+432, lowByte(sumDAC2));
      sumDAC2 = 0;
      if (j == 144) // 144 x 300sec (12 hours)
        {
        j = 0;
        }
      k = k + 1;
      if (k == 144) // 144 x 10800sec (18 days) ?ändra till 160 = 20days?
      {
        k = 0;
      }
      EEPROM.write(1023, k); // store present k (index of 3 hour average, used in setup)
     
    }
  }
  
}

void printDataToSerial()
{
  Serial.print((time/10), DEC);
  Serial.print("\t");              // prints a tab
  Serial.print(timer_us *1000 + TIC_Value , DEC);
  Serial.print("\t"); 
  Serial.print(TIC_Value, DEC);
  Serial.print("\t");              // prints a tab
  Serial.print(TIC_ValueFiltered * 10 / filterConst, DEC);
  Serial.print("\t");              // prints a tab
  Serial.print(dacValueOut, DEC);
  Serial.print("\t");              // prints a tab
  Serial.print(tempADC1, DEC);
  Serial.print("\t");              // prints a tab
  Serial.print(tempADC2, DEC);
  Serial.print("\t");              // prints a tab
/*  Serial.print(modeADC3, DEC);
  Serial.print("\t");              // prints a tab
  Serial.print(timeADC4, DEC);
  Serial.print("\t");              // prints a tab   
  Serial.print(spanADC5, DEC);
  Serial.print("\t");   // prints a tab
*/
  Serial.print(timeConst, DEC);
  Serial.print("\t");              // prints a tab
  Serial.print(timerCountsAcc , DEC);
  Serial.print("\t");              // prints a tab
   Serial.print(timerCountsDiff, DEC);
  Serial.print("\t");              // prints a tab
  Serial.print((missedPPS), DEC);
  Serial.print("\t");              // prints a tab
  if (modeADC3 < 475)
  { 
   Serial.print((PPSlocked), DEC);
   Serial.print("\t");              // prints a tab
  }
  else
    {   
    Serial.print("Hold");
    Serial.print("\t");              // prints a tab
    }
  if (i == 1)
  {
  Serial.print("Five minute averages: TIC+DAC+temp");
  Serial.print("\t");              // prints a tab
  }
  if (i == 2)
  {
  Serial.print("Now acquiring value: ");
  Serial.print(j);
  Serial.print("\t");              // prints a tab
  }
  if ((i >= 4) && (i <= 147))
  {
  Serial.print((i-4), DEC);
  Serial.print("\t");              // prints a tab
  Serial.print((StoreTIC_A[i-4]), DEC);
  Serial.print("\t");              // prints a tab
  Serial.print((StoreDAC_A[i-4]), DEC);
  Serial.print("\t");              // prints a tab
  Serial.print((StoreTempA[i-4]), DEC);
  Serial.print("\t");              // prints a tab  
  }
  if (i == 148)
  {
  Serial.print("Three hour averages: TIC+DAC+temp");
  Serial.print("\t");              // prints a tab
  }
  if (i == 149)
  {
  Serial.print("Now acquiring value: ");
  Serial.print(k);
  Serial.print("\t");              // prints a tab
  }
 if ((i >= 150) && (i <=293))
  {
  Serial.print((i-150+1000), DEC);
  Serial.print("\t");              // prints a tab
  Serial.print((EEPROM.read(i-150+0)*256 + EEPROM.read(i-150+144)), DEC);
  Serial.print("\t");   
  unsigned int x = EEPROM.read(i-150+288)*256 + EEPROM.read(i-150+432); // not nice but...
  Serial.print(x, DEC);
  Serial.print("\t");              // prints a tab   
  Serial.print((EEPROM.read(i-150+576)*256 + EEPROM.read(i-150+720)), DEC);
  Serial.print("\t");              // prints a tab
  } 
 if (i == 295)
  {
  Serial.print("TimeConst = ");
  Serial.print(timeConst);
  Serial.print(" sec ");
  Serial.print("\t");              // prints a tab
  }
 if (i == 296)
  {
  Serial.print("Prefilter = ");
  Serial.print(filterConst);
  Serial.print(" sec ");
  Serial.print("\t");              // prints a tab
  }
 if (i == 297)
  {
  Serial.print("Damping = ");
  Serial.print(damping);
  Serial.print("/100");
  Serial.print("\t");              // prints a tab
  } 
 if (i == 298)
  {
  Serial.print("Arduino GPSDO by Lars Walenius");
  Serial.print("\t");              // prints a tab
  }
 if (i == 299)
  {
  Serial.print("Rev. 20 Feb 12, 2014"); // Could be a string in the beginning of the program
  Serial.print("\t");              // prints a tab
  }
  
  Serial.println("");      // prints a carriage return
  
  PPS_ReadFlag = 0; // Main loop will do nothing before a new PPS interrupt sets this flag again
}

void setup() 
{
  pinMode(ppsLockedLED,OUTPUT);
  pinMode(extraSwitch ,INPUT);
  Serial.begin(9600);
  Serial.println("Arduino GPSDO by Lars Walenius Rev. 20 Feb 12, 2014"); // Could be a string in the beginning of the program
  attachInterrupt(0, pps, RISING);
  k = EEPROM.read (1023); // last index of stored 3 hour average
   if (EEPROM.read(((k+143) % 144) + 288) > 0) // get last stored 3 hour DAC value if not zero (Note: not reverse corrected for temp yet)
    { 
    unsigned int y = EEPROM.read(((k+143) % 144) + 288)*256 + EEPROM.read(((k+143) % 144) + 432); // not nice but...
    dacValueOut = y;
    } 
  valuePWMhigh = highByte(dacValueOut);
  valuePWMlow = lowByte(dacValueOut);
  analogWrite(PWMhighPin,valuePWMhigh);
  analogWrite(PWMlowPin,valuePWMlow);
  dacValue = dacValueOut * timeConst;
  FastLockDacValue = dacValue;
  timeConstOld = timeConst;
  TIC_ValueOld = TIC_Offset * filterConst;
  TIC_ValueFiltered = TIC_Offset * filterConst;
  TIC_ValueFilteredForPPS_lock = TIC_Offset * 16;
  tempADC2_Filtered = tempRef * 100;
  analogReference(INTERNAL);
  
  overflowCount = 0;  // no overflows yet (move to variables?)
// stop Timer 0 so that timer1 not misses counts due to timer0 interrupts
// note: millis() delay()) etc will not work!
  TCCR0A = 0;             
  TCCR0B = 0;  
// reset Timer 1
  TCCR1A = 0;             
  TCCR1B = 0;              
// Timer 1 - counts events on pin D5
// 5MHz external clock source on T1 pin (D5). Clock on rising edge.
  TIMSK1 = _BV (TOIE1);   // interrupt on Timer 1 overflow
  TCNT1 = 0;      //counter 1 to zero
  TCCR1B =  _BV (CS10) | _BV (CS11) | _BV (CS12); // start Timer 1
}

void loop() 
{
  if (PPS_ReadFlag >0)
  {
    calculation();
    printDataToSerial();
  }
}

